```
>
```

Forced Oscillators

```
> restart:with(plots):
```

We consider the vibrations of a spring-mass system when an external force is applied. We choose the interesting case of a periodic forcing function, and consider the following forced oscillator:

$$x'' + .3\, x' + 4.9\, x = 2\cos(4\,t)$$

with initial conditions

x(0)=1 and x'(0)=1.

```
>
```

We first look at the unforced oscillator without the forcing cos term..

```
> unforced:=diff(x(t),t$2)+(3/10)*diff(x(t),t)+(49/10)*x(t)=0;
```

$$unforced := \frac{d^2}{dt^2}\, x(t) + \frac{3}{10}\left(\frac{d}{dt}\, x(t)\right) + \frac{49}{10}\, x(t) = 0$$

We solve the unforced oscillator.

```
> transient:=dsolve(unforced,x(t));
```

$$transient := x(t) = \_C1\, e^{-\frac{3}{20}\,t} \sin\left(\frac{1}{20}\sqrt{1951}\,t\right) + \_C2\, e^{-\frac{3}{20}\,t} \cos\left(\frac{1}{20}\sqrt{1951}\,t\right)$$

We refer to this solution of this underdamped unforced oscillator as the **transient solution** since it approaches 0 as t approaches infinity (i.e., it sort of disappears) because of the exponentials. Now we add in the oscillating forcing function.

```
> forced:=diff(x(t),t$2)+(3/10)*diff(x(t),t)+(49/10)*x(t)=2*cos(4*t)
  ;
```

$$forced := \frac{d^2}{dt^2}\, x(t) + \frac{3}{10}\left(\frac{d}{dt}\, x(t)\right) + \frac{49}{10}\, x(t) = 2\cos(4\,t)$$

We solve the forced oscillator.

```
> solution:=dsolve(forced,x(t));
```

$$solution := x(t) = e^{-\frac{3}{20}\,t} \sin\left(\frac{1}{20}\sqrt{1951}\,t\right) \_C2 + e^{-\frac{3}{20}\,t} \cos\left(\frac{1}{20}\sqrt{1951}\,t\right) \_C1 + \frac{16}{831}\sin(4\,t) - \frac{148}{831}\cos(4\,t)$$

The difference of the solutions of the forced and unforced oscillators is a particular solution. We also refer to this difference as the **steady state solution**.

```
> steadystate:=-148/831*cos(4*t)+16/831*sin(4*t);
```

$$steadystate := \frac{16}{831}\sin(4\,t) - \frac{148}{831}\cos(4\,t)$$

We check that this really is a particular solution.

```
> eval(subs(x(t)=steadystate,forced));
```

$$2\cos(4\,t) = 2\cos(4\,t)$$

The equality indicates that we have a particular solution. Now we solve the forced oscillator with the initial conditions.

```
> IC:=x(0)=1,D(x)(0)=1;
```

$$IC := x(0) = 1, D(x)(0) = 1$$

```
> solutionIC:=dsolve({forced,IC},x(t));
```

$$solutionIC := x(t) = \frac{18277}{1621281} e^{-\frac{3}{20} t} \sin\left(\frac{1}{20}\sqrt{1951}\, t\right)\sqrt{1951} + \frac{979}{831} e^{-\frac{3}{20} t} \cos\left(\frac{1}{20}\sqrt{1951}\, t\right) + \frac{16}{831}\sin(4\, t) - \frac{148}{831}\cos(4\, t)$$

We now plot the solution (blue) along with the steady state (green).

```
> p1:=plot(rhs(solutionIC),t=0..40,color=blue,numpoints=1000):
  p2:=plot(steadystate,t=0..40,color=green,numpoints=1000):
  display(p1,p2);
```



We see that as t increases to infinity and the transient solution goes to 0, the actual solution approaches the steady state solution.

```
>
```